Taylor Series defined recursively

Leo asked after class on Monday whether Taylor Series could be defined recursively. Here they are.

Thanks for asking, Leo: I love talking about Taylor Polynomials!:)

In[139]:=
```
taylor[f_, x_, x0_, n_] :=
 If[n == 0, (* Base
   case: the 0th degree Taylor polynomial is the function evaluated at x0: *)
  f[x0],
  (* Inductive step:
    the nth degree Taylor polynomial is the (n-1)th Taylor polynomial,
  plus one more term: f^(n)(x0) (x-x0)^n/n!: *)
  Module[{fn},
   fn[y_] = D[f[y], {y, n}]; (* the nth derivative function.
     If I try to use "x" for the variable in the definition of fn,
   Mathematica complains because it's unhappy about me using an x
      within the scope of something which already uses an x....*)
   taylor[f, x, x0, n - 1](* the previous Taylor polynomial *)
    + fn[x0] (x - x0) ^ n / Factorial[n] (* the new term. *)
  ]
 ]
```

Let’s use polynomials to approximate the Sine function on the interval $[0, \frac{\pi}{2}]$ (which is exactly what your calculator does, by the way):

In[140]:=
```
f[x_] = Sin[x]
x0 = Pi / 4 (* we'll expand about the midpoint of the interval *)
```

Out[140]= Sin[x]

Out[141]= $\frac{\pi}{4}$

Degree 0: gets the function exactly right at $x_0 = \frac{\pi}{4}$:

In[142]:=
```
taylorConstant = taylor[f, x, x0, 0]
Plot[{f[x], taylorConstant}, {x, 0, Pi / 2}, PlotLabels → Automatic]
```

Out[142]= $\frac{1}{\sqrt{2}}$

Out[143]=



Degree 1: Tangent line approximation at $x_0 = \frac{\pi}{4}$:

In[144]:=
```
taylorLinear = taylor[f, x, x0, 1]
Plot[{f[x], taylorLinear}, {x, 0, Pi / 2}, PlotLabels → Automatic]
```

Out[144]= $\frac{1}{\sqrt{2}} + \frac{-\frac{\pi}{4} + x}{\sqrt{2}}$

Out[145]=



Degree 2: Tangent Quadratic approximation

In[146]:=
```
taylorQuadratic = taylor[f, x, x0, 2]
Plot[{f[x], taylorQuadratic}, {x, 0, Pi / 2}, PlotLabels → Automatic]
```

Out[146]= $\frac{1}{\sqrt{2}} + \frac{-\frac{\pi}{4} + x}{\sqrt{2}} - \frac{\left(-\frac{\pi}{4} + x\right)^2}{2\sqrt{2}}$

Out[147]=



Degree 3: Tangent Cubic approximation

In[148]:=
```
taylorCubic = taylor[f, x, x0, 3]
Plot[{f[x], taylorCubic}, {x, 0, Pi / 2}]
```

Out[148]= $\frac{1}{\sqrt{2}} + \frac{-\frac{\pi}{4} + x}{\sqrt{2}} - \frac{\left(-\frac{\pi}{4} + x\right)^2}{2\sqrt{2}} - \frac{\left(-\frac{\pi}{4} + x\right)^3}{6\sqrt{2}}$

Out[149]=



Degree 4: Tangent Quartic approximation

In[150]:=
```
taylorQuartic = taylor[f, x, x0, 4]
Plot[{f[x], taylorQuartic}, {x, 0, Pi / 2}]
```

Out[150]= $\frac{1}{\sqrt{2}} + \frac{-\frac{\pi}{4} + x}{\sqrt{2}} - \frac{\left(-\frac{\pi}{4} + x\right)^2}{2\sqrt{2}} - \frac{\left(-\frac{\pi}{4} + x\right)^3}{6\sqrt{2}} + \frac{\left(-\frac{\pi}{4} + x\right)^4}{24\sqrt{2}}$

Out[151]=



Degree 5: Tangent Quintic approximation

In[152]:=
```
taylorQuintic = taylor[f, x, x0, 5]
Plot[{f[x], taylorQuintic}, {x, 0, Pi / 2}, PlotLabel → "Any difference?"]
Plot[{f[x] - taylorQuintic}, {x, 0, Pi / 2},
 PlotLabel → "Error: Difference between f and the quintic"]
```

Out[152]=
$$\frac{1}{\sqrt{2}} + \frac{-\frac{\pi}{4} + x}{\sqrt{2}} - \frac{\left(-\frac{\pi}{4} + x\right)^2}{2\sqrt{2}} - \frac{\left(-\frac{\pi}{4} + x\right)^3}{6\sqrt{2}} + \frac{\left(-\frac{\pi}{4} + x\right)^4}{24\sqrt{2}} + \frac{\left(-\frac{\pi}{4} + x\right)^5}{120\sqrt{2}}$$

Out[153]=


Out[154]=


By the way, the **form** of each of the expressions for the Taylor polynomials above is a **bad way to evaluate them**.

Why? And how could we improve the expressions?